

Optimización tecno-económica de proyectos de autogeneración AGPE con PV y BESS



PAULO M. DE OLIVEIRA-DE JESUS

<https://power.uniandes.edu.co/people/professor/pdeoliveira>

Ciclo de 6 semanas – 12 sesiones

Módulo 4 – Clase 8 - 28 de mayo de 2026



ESPECIFICACION DEL SISTEMA PV/BESS

CRITERIOS España: Autoconsumo en general


- A) Solo PV Sin BESS: Mercado Spot**
- B) PV & BESS: Mercado Spot**

ESPECIFICACION DEL SISTEMA PV/BESS

CRITERIOS Colombia (AGPE < 1000 kW)

- A) Solo PV Sin BESS: Sólo crédito de energía (Exc 1),
- B) Solo PV Sin BESS: Crédito + Excedentes horarios (Exc2)
- C) PV & BESS: Solo crédito, Exc 2 = 0 (***Optimo***)
- D) PV & BESS: Crédito + Exc 2 > 0 (***Optimo***)

Gestión Óptima de Autogeneración Industrial con FNCER y Almacenamiento



Conociendo los precios spot DAY-AHEAD y la previsión de radiación horaria para el día siguiente:

¿cómo **gestionar la batería** y la **importación/exportación en la frontera** para **maximizar** la valorización de excedentes atendiendo una **carga inflexible**?

Figura 18: Autogeneración Industrial con FNCER y BESS


El modelo de optimización procura maximizar el beneficio económico operacional del intercambio horario de energía con la red de servicio público, importaciones en kW P_b^t y exportaciones en kW P_c^t , parámetros que son dependientes de la gestión de la batería en el tiempo, carga en kW P_c^t , descarga en kW P_d^t y el estado de carga SOC^t en kWh.

El objetivo del *problema de optimización* es determinar el patrón de carga y descarga de las baterías así como determinar el esquema de compra-venta de energía con la red tal que se maximice el beneficio de la operación, tomando la posible ventaja de la variabilidad de los precios durante el día. Se incorporan variables binarias que impiden de la batería cargue y descargue de forma simultánea.

Los modelos son específicos para cada mercado

Conociendo los precios spot
DAY-AHEAD y la previsión de
radiación horaria para el día
siguiente:

¿cómo **gestionar la batería** y
la **importación/exportación en**
la frontera para
maximizar la valorización de
excedentes atendiendo
una *carga inflexible*?



Modelo de 24 horas

Modelo de 1x8760 o 365x24 horas es útil para realizar análisis forense:
¿cual hubiese sido el resultado de la valoración de excedentes si se
hubiese optimizado la operación?



Modelo Gestión Óptima de AG con PV y BESS - España



España

sujeto a:

El modelo matemático del despacho se describe a continuación:

Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

$$\max_{\mathbb{P}_s, \mathbb{P}_b, \mathbb{P}_c, \mathbb{P}_d, \text{SOC}, \mathbf{w}, \mathbf{u}} \text{Beneficio} = -CP + 365 \sum_{t=1}^{T=24} \lambda^t P_s^t - (\lambda^t + \psi^t) P_b^t$$

$$P_d^t + P_b^t + P_{pv}^{u,t} P_{pv}^{max} = P_c^t + P_s^t + P_L^{u,t} P_L^{max}, \forall t$$

$$P_c^t \leq R_c C w^t, \forall t$$

$$P_d^t \leq R_d C (1 - w^t), \forall t$$

$$P_b^t \leq P_{max} u^t, \forall t$$

$$P_s^t \leq P_{max} (1 - u^t), \forall t$$

$$SOC^1 = SOC^0 + P_c^1 \eta_c - \frac{P_d^1}{\eta_d}$$

$$SOC^t = SOC^{t-1} + P_c^t \eta_c - \frac{P_d^t}{\eta_d}, \forall t > 1$$

$$\beta_{min} C \leq SOC^t \leq \beta_{max} C, \forall t$$

$$SOC^T = SOC^0$$

$$P_b^t \leq P_{b,p}^{max}, \forall t$$

MODELO MILP

Mixed-Integer Linear Programming

Donde:

$\mathbb{P}_s = [P_s^1, \dots, P_s^T]$; $\mathbb{P}_b = [P_b^1, \dots, P_b^T]$; $\mathbb{P}_c = [P_c^1, \dots, P_c^T]$; $\mathbb{P}_d = [P_d^1, \dots, P_d^T]$; $\text{SOC} = [SOC^1, \dots, SOC^T]$;

$\mathbf{w} = [w^1, \dots, w^T]$; $\mathbf{u} = [u^1, \dots, u^T]$ son los conjuntos que contienen las variables de estado

CP es el cargo por potencia en Eur/año,



España

sujeto a:

El modelo matemático del despacho se describe a continuación:

$$\max_{P_s, P_b, P_c, P_d, SOC, w, u} \text{Beneficio} = -CP + 365 \sum_{t=1}^{T=24} \lambda^t P_s^t - (\lambda^t + \psi^t) P_b^t \quad [\text{Eur/Año}]$$
$$CP = \sum_{p=1}^6 \kappa_p P_{b,p}^{max}$$

CP es el cargo por potencia en Eur/año,

No se optimiza $P_{b,p}^{max}$

Pero es interesante ver el impacto de renegociar los contratos de potencia



España

sujeto a:

El modelo matemático del despacho se describe a continuación:

$$\text{Beneficio} = -CP + 365 \sum_{t=1}^{T=24} \lambda^t P_s^t - (\lambda^t + \psi^t) P_b^t \quad [\text{Eur/Año}]$$

Diagram illustrating the mathematical model for dispatch scheduling:

- Variables (pointing to the equation):**
 - $P_s, P_b, P_c, P_d, SOC, w, u$ (pointing to the maximization operator máx)
 - P_s (pointing to $\lambda^t P_s^t$)
 - P_b (pointing to $(\lambda^t + \psi^t) P_b^t$)
- Interpretations (pointing to the terms in the equation):**
 - λ^t : Precio Spot
 - ψ^t : peaje energía
 - P_s^t : Ventas
 - P_b^t : Compras
- Operational States (pointing to the variables w, u):**
 - Estado de Carga
 - Descarga de BESS
 - Carga de BESS
 - Compras
 - Ventas

Cada variable en AZUL es un conjunto de 24 incógnitas

Las variables en verde son DATOS



España

sujeto a:

$$P_d^t + P_b^t + P_{pv}^{u,t} P_{pv}^{max} = P_c^t + P_s^t + P_L^{u,t} P_L^{max}, \forall t$$

Balance de Energía Horaria

$$P_c^t \leq R_c C w^t, \forall t$$

Carga de batería no debe exceder su capacidad C-rate x Capacidad

$$P_d^t \leq R_d C (1 - w^t), \forall t$$

Descarga de batería no debe exceder su capacidad C-rate x Capacidad

$$P_b^t \leq P_{max} u^t, \forall t$$

Importación no debe exceder su capacidad de la frontera

$$P_s^t \leq P_{max} (1 - u^t), \forall t$$

Exportación no debe exceder su capacidad de la frontera

$$SOC^1 = SOC^0 + P_c^1 \eta_c - \frac{P_d^1}{\eta_d}$$

$$SOC^t = SOC^{t-1} + P_c^t \eta_c - \frac{P_d^t}{\eta_d}, \forall t > 1$$

BALANCE en la Batería
considerando las eficiencias de
carga y descarga

$$\beta_{min} C \leq SOC^t \leq \beta_{max} C, \forall t$$

Estado de carga limitado a un mínimo y máximo
dependiendo de la profundidad de descarga permitida

$$SOC^T = SOC^0$$

Ciclo simétrico

$$P_b^t \leq P_{b,p}^{max}, \forall t$$

Importación no debe exceder su capacidad contratada en
cada bloque horario

w^t y u^t son variables de decisión binarias.



España

Ejemplo 5 - Autogeneración PV & BESS - España



Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

```
GAMS Studio - No active process

Welcome | M1_4_E5e.gms

1 $title Despacho óptimo Industria+BESS+PV OnGrid (Caso España)
2 *M1.4.Ej5e - Caso España: PV+BESS (OnGrid) Despacho Optimo
3 Set
4   t 'hours' / t1*t24 /;
5 Table data(t,*)
6   * Eur/kWh      Eur/kWh      per unit      per unit      kW
7   Lambda        psi          Ppvu          Plu          Pbmax
8   t1 0.068746794520548 0.0228205053080510 0.000000000000000 0.135388629090637 564.30000000000
9   t2 0.064382931506849 0.0228205053080510 0.000000000000000 0.090377249176867 564.30000000000
10  t3 0.062238410958904 0.0228205053080510 0.000000000000000 0.076343641406536 564.30000000000
11  t4 0.060464136986301 0.0228205053080510 0.000000000000000 0.075224251241957 564.30000000000
12  t5 0.060655534246575 0.0228205053080510 0.000000000000000 0.074307578750070 564.30000000000
13  t6 0.065132136986301 0.0228205053080510 0.000000000000000 0.073105608680250 564.30000000000
14  t7 0.073826082191781 0.0228205053080510 0.047492750000000 0.073406190057000 564.30000000000
15  t8 0.076148794520548 0.0228205053080510 0.200034333333333 0.085516408757498 564.30000000000
16  t9 0.069382712328767 0.0210743713052406 0.371942916666667 0.883664344082149 338.05515300000
17  t10 0.056760082191781 0.0319883482993510 0.512186250000000 1.000000000000000 404.47000000000
18  t11 0.046949369863014 0.0221906474349675 0.605126833333333 0.967245871300936 442.91000000000
19  t12 0.041846739726027 0.0221906474349675 0.645309833333333 0.966410238486499 442.91000000000
20  t13 0.039593726027397 0.0221906474349675 0.627257666666667 0.122847383567665 442.91000000000
21  t14 0.037885123287671 0.0221906474349675 0.568799916666667 0.892179315688642 442.91000000000
22  t15 0.036840931506849 0.0210743713052406 0.473205250000000 0.968865480059381 338.05515300000
23  t16 0.038253698630137 0.0210743713052406 0.337161416666667 0.959187873431187 338.05515300000
24  t17 0.046423205479452 0.0210743713052406 0.188570083333333 0.947290206165413 338.05515300000
25  t18 0.059484767123288 0.0210743713052406 0.043513583333333 0.647289021374806 338.05515300000
26  t19 0.076155178082192 0.0720344854462120 0.000000000000000 0.627968640962237 504.43000000000
27  t20 0.090286438356164 0.0512182441617940 0.000000000000000 0.659335616819762 453.96686700000
28  t21 0.099023506849315 0.0512182441617940 0.000000000000000 0.581590012689108 453.96686700000
29  t22 0.090372109589041 0.0512182441617940 0.000000000000000 0.179217943416770 453.96686700000
30  t23 0.081188794520548 0.0210743713052406 0.000000000000000 0.144883067091137 338.05515300000
31  t24 0.074767232876712 0.0210743713052406 0.000000000000000 0.138041019820362 338.05515300000;
```



España

Ejemplo 5 - Autogeneración PV & BESS - España

```
33 PARAMETER
34 pbmax(t);
35 pbmax(t) = data(t, 'Pbmax');
36 Variable
37     Benefit      'Benefit Eur/year'
38     Capacity     'Capacity Payment Eur/year'
39     Energy       'Energy Trading Result Eur/year'
40     CashFlow     'Project CashFlow Eur/yr'
41     npv          'Net Present Value Eur'
42     Base         'Load payment with no PV and no BESS';
43 Positive variable
44     SOC(t)       'State of Charge kWh'
45     Pd(t)        'Power battery discharge kW'
46     Pc(t)        'Power battery charge kW'
47     Pb(t)        'Power bought from the public grid kW'
48     Ps(t)        'Power sold to the public grid kW'
49     Investment   'Project initial investment Eur'
50     Wload        'Energia total load kWh/año'
51     Wpvtot       'Producción Energia PV kWh/año'
52     Wimport      'Importación Energia kWh/año'
53     Wexport      'Exportación Energia kWh/año'
54     WcargaBESS   'Energia carga BESS kWh/año'
55     WdescargaBESS 'Energia carga BESS kWh/año';
56 Binary Variable
57     w1(t)
58     w3(t);
```



España

Ejemplo 5 - Autogeneración PV & BESS - España



Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

```
59  Scalar
60  Ppvmax      / 569/
61  Plmax       / 231.241369863014 /
62  SOC0        / 100 /
63  C           / 2000 /
64  eff_c       / 0.93 /
65  eff_d       / 0.97 /
66  R_c         / 0.2/
67  R_d         / 0.2/
68  DoD         / 0.9/
69  kappaP1     / 28.79187 /
70  kappaP2     / 15.07764 /
71  kappaP3     / 6.55917 /
72  kappaP4     / 5.17209 /
73  kappaP5     / 1.93281 /
74  kappaP6     / 0.91609 /
75  Pmax        / 600/
76  CAPEXpv     /700/
77  CAPEX_BESS  /200/
78  CAPEX_BESS_inverter /80/
79  crf         /0.08024258719069/
80  Capacity0   /-31713.9/;
81  SOC.up(t)    = ((1-DoD)/2+DoD)*C;
82  SOC.lo(t)    = ((1-DoD)/2)*C;
83  SOC.fx('t24') = SOC0;
```




España

Ejemplo 5 - Autogeneración PV & BESS - España



Universidad de
los Andes

Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

```

84 Equation Bcalc, Bcalc1, Bcalc2, balance, r1, r2, r3, r4, r5, r6, r7, r8, r9, r10, r11, r12, r13, r14,
85 Bcalc..      Benefit =e= Capacity+Energy;
86 Bcalc1..     Capacity =e= -kappaP1*pbmax('t19')-kappaP2*pbmax('t20')-kappaP3*pbmax('t10')-kappaP4*pbmax('t11');
87 Bcalc2..     Energy =e=365*sum((t), (data(t,'lambda')*ps(t)-(data(t,'lambda')+data(t,'psi'))*pbmax(t)));
88 balance(t).. Pd(t) + Pb(t) + Ppvmax*data(t,'Ppvu') - Pc(t) - Ps(t) - Plmax*data(t,'Plu') =e= 0;
89 r1(t)..      SOC(t) =e= SOC0$(ord(t)=1) + SOC(t-1)$(ord(t)>1) + Pc(t)*eff_c - Pd(t)/eff_d;
90 r2(t)..      Pc(t) =l= R_c*C*w1(t);
91 r3(t)..      Pd(t) =l= R_d*C*(1-w1(t));
92 r4(t)..      Pb(t) =l= Pmax*w3(t);
93 r5(t)..      Ps(t) =l= Pmax*(1-w3(t));
94 r6(t)..      pbmax(t) =g= Pb(t);
95 r7..         CashFlow =e= Benefit-Base;
96 r8..         Investment =e= CAPEXpv*Ppvmax+CAPEX_BESS*C+CAPEX_BESS_inverter*C*R_c;
97 r9..         npv=e=CashFlow/crf-Investment;
98 r10..        Base =e=365*sum((t), -(data(t,'lambda')+data(t,'psi'))* Plmax*data(t,'Plu')+Capacity0;
99 r11..        Wload=e=365*sum((t),data(t,'Plu'))*Plmax;
100 r12..        Wpvtot=e=365*sum((t),data(t,'Ppvu'))*Ppvmax;
101 r13..        Wimport=e= 365*sum((t),Pb(t));
102 r14..        Wexport=e= 365*sum((t),Ps(t));
103 r15..        WcargaBESS =e= 365*sum((t),Pc(t));
104 r16..        WdescargaBESS =e= 365*sum((t),Pd(t));
105 *option mip=gurobi minlp=gurobi, miqcp=gurobi;
106 *option mip=lindo;
107 *option mip=clpex;
108 Model modelAGspain / all /;
109 solve modelAGspain using mip maximizing Benefit
110 execute_unload 'Results M1_4_E5e.gdx';

```



España

Ejemplo 5 - Autogeneración PV & BESS - España




Figura 19: Balance anual de energía - Ejemplo 5 (España)

Las importaciones (compras) fueron -35353.1 Eur/año y las exportaciones (ventas de excedentes) fue 13614.6 Eur/año). El balance fue positivo 7235.7 Eur/año (exportador neto).



España

Ejemplo 5 - Autogeneración PV & BESS - España



Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

Potencia contratada (no cambia)

Período	Término de Potencia κ_p (Eur/kW-año)	$P_{b,p}^{max}$ (kW)	CP (Eur)
P1	28.791866	504.4	14523.5
P2	15.077643	454.0	6844.8
P3	6.559170	404.5	2653.0
P4	5.172085	442.9	2290.8
P5	1.932805	338.1	516.9
P6	0.916088	564.3	339.0
		Total	27482.3



España

Ejemplo 5 - Autogeneración PV & BESS - España

¡ÓPTIMO!



	Sin PV (Eur/año)	Con PV (Eur/año)	Ahorro (Eur/año)
Energia Activa	- 82387.75	7235.7	89623.4
Termino Potencia	-31713.89	-27482.33	4231.56
Total	- 114101.64	-20246.6	93855.0

Tabla 14: Costo energético de una industria genérica en España con AG 569 kWp y BESS 2000 kWh, 2024

AHORRO de 82%



España

Evaluación Económica

NPV

TIR

PBT

B/C Ratio

VP Sistema PV a 700 Eur/kWp


VP Sistema BESS: 200 Eur/kWh (batería)

80 Eur/kW (inversor GFL)

(incluye costo inversor y accesorios)

Tasa de descuento 5%

Vida del proyecto 20 años





España

Ejemplo 5 - Autogeneración PV & BESS - España



Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

Código	Tecn.	¿Desp. óptimo?	VPN (Eur)	TIR (%)	PBT (años)	B/C	Conf.
M1.4.E3e	PV	no	394710	15.0	7.7	2.0	Media
M1.4.E5e	PV+BESS	si	339341	9.4	12	1.4	Alta

Tabla 15: Indicadores de bondad financiera del proyecto PV+BESS en España, 2024

¡TIR solo PV mejor que TIR PV+BESS!

El costo de la batería (sigue muy alto) penaliza el proyecto.

No obstante tener batería mejora confiabilidad


¿cuanto cuesta la confiabilidad?



España

Ejemplo 5 - Autogeneración PV & BESS - España

DESPACHO OPTIMO





España

Ejemplo 5 - Autogeneración PV & BESS - España

DESPACHO OPTIMO






España

Ejemplo 5 - Autogeneración PV & BESS - España

DESPACHO OPTIMO





España

Ejemplo 5 - Autogeneración PV & BESS - España

DESPACHO OPTIMO





Colombia



Colombia

Modelo Gestión Optima de AGPE con PV y BESS - Colombia

Facultad de
Ingeniería

Educación
Continua
Vicerrectoría Académica

El modelo matemático del despacho se describe a continuación:

$$\max_{\mathbb{P}_s, \mathbb{P}_b, \mathbb{P}_c, \mathbb{P}_d, \text{SOC}, \mathbf{w}, \mathbf{u}} \text{Beneficio} = -\text{Backup} + VE \quad [\text{USD/Año}]$$

$$\text{Backup} = 12\kappa P_b^{\max}$$

$$VE = 365[(\text{Exc}_1 - \text{Imp})c_u - \text{Exc}_1 C_v(1 - \gamma) - \text{Exc}_1(T + D + P + C_v + R)\gamma + \text{Exc}_2]MC_m]$$

$$\text{Imp} = 365 \sum_{t=1}^{24} P_b^t \quad \text{Export} = 365 \sum_{t=1}^{24} P_s^t \quad \text{Exc}_1 = \text{Export} \times w_2 + \text{Imp} \times (1 - w_2) \quad \text{Exc}_2 = 0 \times w_2 + (\text{Export} - \text{Imp}) \times (1 - w_2)$$

sujeto a:

$$P_d^t + P_b^t + P_{pv}^{u,t} P_{pv}^{\max} = P_c^t + P_s^t + P_L^{u,t} P_L^{\max}, \forall t$$

$$P_c^t \leq R_c C w^t, \forall t$$

$$P_d^t \leq R_d C(1 - w^t), \forall t$$

$$P_b^t \leq P_{\max} u^t, \forall t$$

$$P_s^t \leq P_{\max}(1 - u^t), \forall t$$

$$\text{SOC}^1 = \text{SOC}^0 + P_c^1 \eta_c - \frac{P_d^1}{\eta_d}$$

$$\text{SOC}^t = \text{SOC}^{t-1} + P_c^t \eta_c - \frac{P_d^t}{\eta_d}, \forall t > 1$$

$$\beta_{\min} C \leq \text{SOC}^t \leq \beta_{\max} C, \forall t$$

$$\text{SOC}^T = \text{SOC}^0$$

$$\text{Exc}_1 \leq \text{Imp}$$

$$P_s^t \leq P_b^{\max}$$

Donde:

$$\mathbb{P}_s = [P_s^1, \dots, P_s^T]; \mathbb{P}_b = [P_b^1, \dots, P_b^T]; \mathbb{P}_c = [P_c^1, \dots, P_c^T]; \mathbb{P}_d = [P_d^1, \dots, P_d^T]; \text{SOC} = [\text{SOC}^1, \dots, \text{SOC}^T];$$

$\mathbf{w} = [w^1, \dots, w^T]; \mathbf{u} = [u^1, \dots, u^T]$ son los conjuntos que contienen las variables de estado



Modelo Gestión Óptima de AGPE con PV y BESS - Colombia

El modelo matemático del despacho se describe a continuación:

$$\max_{P_s, P_b, P_c, P_d, SOC, w, u} \text{Beneficio} = -Backup + VE \quad [\text{USD/Año}]$$



$$Backup = 12\kappa P_b^{max}$$

CONTRATO DE RESPALDO SEGUN CREG 015/2018

P_b^{max} no se optimiza



Modelo Gestión Optima de AGPE con PV y BESS - Colombia

El modelo matemático del despacho se describe a continuación:

$$\max_{P_s, P_b, P_c, P_d, SOC, w, u} \text{Beneficio} = -Backup + VE \quad [\text{USD/Año}]$$

↑ ↑ ↑ ↑ ↑
Ventas Compras Carga de BESS Descarga de BESS Estado de Carga


↓
Valorización de Excedentes



Colombia

Valorización de Excedentes [USD/Año]

$$VE = 365[(Exc_1 - Imp)c_u - Exc_1 C_v(1 - \gamma) - Exc_1(T + D + P + C_v + R)\gamma + Exc_2]MC_m$$



Si AGPE es mayor a 100 kW γ es 1
Si AGPE es menor a 100 kW γ es 0

$w_2 = 0$ si se permiten excedentes
TIPO 2
 $w_2 = 1$ no se permiten excedentes
TIPO 2



Colombia

sujeto a:

$$P_d^t + P_b^t + P_{pv}^{u,t} P_{pv}^{max} = P_c^t + P_s^t + P_L^{u,t} P_L^{max}, \forall t$$

Balance de Energía Horaria

$$P_c^t \leq R_c C w^t, \forall t$$

Carga de batería no debe exceder su capacidad C-rate x Capacidad

$$P_d^t \leq R_d C (1 - w^t), \forall t$$

Descarga de batería no debe exceder su capacidad C-rate x Capacidad

$$P_b^t \leq P_{max} u^t, \forall t$$

Importación no debe exceder su capacidad de la frontera

$$P_s^t \leq P_{max} (1 - u^t), \forall t$$

Exportación no debe exceder su capacidad de la frontera

$$SOC^1 = SOC^0 + P_c^1 \eta_c - \frac{P_d^1}{\eta_d}$$

$$SOC^t = SOC^{t-1} + P_c^t \eta_c - \frac{P_d^t}{\eta_d}, \forall t > 1$$

BALANCE en la Batería
considerando las eficiencias de
carga y descarga

$$\beta_{min} C \leq SOC^t \leq \beta_{max} C, \forall t$$

Estado de carga limitado a un mínimo y máximo
dependiendo de la profundidad de descarga permitida

$$SOC^T = SOC^0$$

Ciclo simétrico

$$P_b^t \leq P_{b,p}^{max}, \forall t$$

Importación no debe exceder su capacidad contratada en
cada bloque horario

w^t y u^t son variables de decisión binarias.



Ejemplo 6 - Autogeneración PV & BESS - Colombia

```

1 $title Despacho óptimo Industria+BESS+PV OnGrid (Caso Colombia ,24H)
2 Set
3   t 'hours'          / t1*t24 /;
4
5 Table data(t,*)
6   *      per unit      per unit      USD/kWh (XM 2024)
7   Plu      Ppvu      lambda
8   t1      0.135388629090637  0.000000000000000  0.1586041451975490
9   t2      0.090377249176867  0.000000000000000  0.1217198153796440
10  t3      0.076343641406536  0.000000000000000  0.1190982516810140
11  t4      0.075224251241957  0.000000000000000  0.1160118975714240
12  t5      0.074307578750070  0.000000000000000  0.1184892975714240
13  t6      0.073105608680250  0.000000000000000  0.1242799509960820
14  t7      0.073406190057000  0.047492750000000  0.1243378325029310
15  t8      0.085516408757498  0.200034333333333  0.1306424242837530
16  t9      0.883664344082149  0.371942916666667  0.1376351153796440
17  t10     1.000000000000000  0.512186250000000  0.1401095797632050
18  t11     0.967245871300936  0.605126833333333  0.1443161345577260
19  t12     0.966410238486499  0.645309833333333  0.1478920468864930
20  t13     0.122847383567665  0.627257666666667  0.1479728318180000
21  t14     0.892179315688642  0.568799916666667  0.1485314667495070
22  t15     0.968865480059381  0.473205250000000  0.1513679722289590
23  t16     0.959187873431187  0.337161416666667  0.1522125852426570
24  t17     0.947290206165413  0.188570083333333  0.1506114811330680
25  t18     0.647289021374806  0.043513583333333  0.1484249605851230
26  t19     0.627968640962237  0.000000000000000  0.1589462482563560
27  t20     0.659335616819762  0.000000000000000  0.1631816996262190
28  t21     0.581590012689108  0.000000000000000  0.1568946133248490
29  t22     0.179217943416770  0.000000000000000  0.1483866982563560
30  t23     0.144883067091137  0.000000000000000  0.1400058133248490
31  t24     0.138041019820362  0.000000000000000  0.1345652879823840;
32 * -----

```



Ejemplo 6 - Autogeneración PV & BESS - Colombia

```

34 Variable
35 Benefit 'Benefit USD/yr'
36 VE 'Energy sold to the system USD/day'
37 CashFlow 'Project CashFlow USD/yr'
38 npv 'Net Present Value USD'
39 Base 'Load payment with no PV and no BESS, USD';
40 Positive variable
41 Backup 'Backup System Payment in USD/kW-yr'
42 SOC(t) 'State of Charge kWh'
43 Pd(t) 'Power battery discharge kW'
44 Pc(t) 'Power battery charge kW'
45 Pb(t) 'Power bought from the public grid kW'
46 Ps(t) 'Power sold to the public grid kW'
47 Investment 'Project initial investment Eur'
48 Exc1 'Exported energy Type 1 (Energy Credit) USD/day'
49 Exc2 'Exported energy Type 2 (Traded with the market) USD/day'
50 Imp 'Imported energy from the system USD/day'
51 expo 'Exported energy kWh/yr'
52 Wload 'Energía total load kWh/año'
53 Wpvtot 'Producción Energía PV kWh/año'
54 Wimport 'Importación Energía kWh/año'
55 Wexport 'Exportación Energía kWh/año'
56 WcargaBESS 'Energía carga BESS kWh/año'
57 WdescargaBESS 'Energía carga BESS kWh/año';
58 Binary Variable
59 w1(t) 'defines if the BESS is charging or discharging'
60 * w2 'defines if Exp < Imp (w2=1) or Exp > Imp (w2=0)'
61 w3(t) 'defines if the AGPE is exporting or importing energy from the system';

```



Ejemplo 6 - Autogeneración PV & BESS - Colombia

```
62  Scalar
63  Ppvmax      / 569/
64  Plmax      / 231.241369863014 /
65  SOC0       / 100 /
66  C          / 2000 /
67  eff_c      / 0.93 /
68  eff_d      / 0.97 /
69  R_c        / 0.2/
70  R_d        / 0.2/
71  DoD        / 0.9/
72  Pmax       / 600 /
73  Pbmax      / 400/
74  Tr         / 0.013405417/
75  Di         / 0.027124375/
76  PR         / 0.012440625/
77  Co         / 0.033245000/
78  Re         / 0.004844167/
79  CU         / 0.190428958/
80  kappa      / 5/
81  Flag       / 1/
82  Cm         /0.1544392817773160/
83  CAPEXpv    /770/
84  CAPEX_BESS /220/
85  CAPEX_BESS_inverter /88/
86  ReactivePayment /-24942.4/
87  crf        /0.08024258719069/
88  w2         /1/ ;
89  SOC.up(t)   = ((1-DoD)/2+DoD)*C;
90  SOC.lo(t)   = ((1-DoD)/2)*C;
91  SOC.fx('t24') = SOC0;
```

$w_2 = 1$
Se obliga al modelo
a que $\text{Exc2} = 0$



Ejemplo 6 - Autogeneración PV & BESS - Colombia

```

92 Equation AGPE, balance, r1, r2, r3, r4a, r4b, r5, r6, r7, r8, r9, r10, r11, r12, r13, r14, r15, r16,
93 AGPE..      Benefit =e= -Backup + VE;
94 r1..        Backup  =e= 12*kappa*Pbmax;
95 r2..        Imp     =e= 365*sum((t),Pb(t));
96 r3..        expo    =e= 365*sum((t),Ps(t));
97 r4a..       Exc1    =e= w2*expo+(1-w2)*Imp;
98 r4b..       Exc2    =e= w2*0+(1-w2)*(expo-Imp);
99 r5..        VE      =e= ((Exc1-Imp)*CU-Exc1*Co*(1-Flag)-Exc1*(Tr+Di+PR+Co+Re)*Flag+ Exc2*Cm);
100 balance(t).. Pd(t) + Pb(t) + Ppvmax*data(t,'Ppvu') - Pc(t) - Ps(t) - Plmax*data(t,'Plu') =e= 0;
101 r6(t)..      SOC(t) =e= SOC0$(ord(t)=1) + SOC(t-1)$ (ord(t)>1) + Pc(t)*eff_c - Pd(t)/eff_d;
102 r7(t)..      Pc(t)  =l= R_c*C*w1(t);
103 r8(t)..      Pd(t)  =l= R_d*C*(1-w1(t));
104 r9(t)..      Pb(t)  =l= Pmax*w3(t);
105 r10(t)..     Ps(t)  =l= Pmax*(1-w3(t));
106 r11..        Exc1   =l= Imp;
107 r12(t)..     Ps(t)  =l= Pbmax;
108 r13..        CashFlow =e= Benefit-Base;
109 r14..        Investment =e= CAPEXpv*Ppvmax+CAPEX_BESS*C+CAPEX_BESS_inverter*C*R_c;
110 r15..        npv=e=CashFlow/crf-Investment;
111 r16..        Base =e=365*sum((t), -CU* Plmax*data(t,'Plu'))+ReactivePayment;
112 r17..        Wload=e=365*sum((t),data(t,'Plu'))*Plmax;
113 r18..        Wpvtot=e=365*sum((t),data(t,'Ppvu'))*Ppvmax;
114 r19..        Wimport=e= 365*sum((t),Pb(t));
115 r20..        Wexport=e= 365*sum((t),Ps(t));
116 r21..        WcargaBESS =e= 365*sum((t),Pc(t));
117 r22..        WdescargaBESS =e= 365*sum((t),Pd(t));
118 option mip=gurobi, minlp=gurobi, miqcp=gurobi;
119 Model modelAGPEcolombia / all /;
120 solve modelAGPEcolombia using mip maximizing benefit
121 execute_unload 'Results M1_4_E6c.gdx';

```



Colombia

Ejemplo 6 - Autogeneración PV & BESS - Colombia

Se obtienen los siguientes resultados técnico-económicos para el caso en que la industria que con actividad de autogeneración/almacenamiento maximice su beneficio en condición de importador neto ($w_2=1$).

El balance energético anual detallado (Fig. 21) se muestra a continuación:

Carga industrial: 959636.7 kWh

Producción Solar: 959629.5 kWh

Importaciones ($Imp=365\sum_{t=1}^{24} P_b^t$): -44290.30 kWh

Exportaciones ($Expo=365\sum_{t=1}^{24} P_s^t$): 12369.34 kWh

Carga de la Bateria: -325982.51 kWh

Descarga de la Batería: 294069.00 kWh

Balance: 0 kW

Los excedentes tipo 1 (Exc_1) corresponden a la exportación total, 12369.34 kWh. Los excedentes tipo 2 (Exc_2) son nulos:

Excedentes Tipo 1 ($Exc_1=Expo$, $Expo < Imp$): 12369.34 kWh/año

Excedentes Tipo 2 ($Exc_2=0$, $Expo < Imp$): 0 kWh/año

$w_2=1$
Se obliga al modelo
 $Exc_2 = 0$



Ejemplo 6 - Autogeneración PV & BESS - Colombia




Figura 21: Balance anual de energía - Ejemplo 6 (Colombia)



Colombia

Ejemplo 6 - Autogeneración PV & BESS - Colombia

La aplicación del Art. 26 de la CREG 174-2021 (modificada en CREG 101 072-2025) produce el siguiente resultado:

Parte1: $(Exc_1 - Imp) \times CU = -6078.7$ US\$/año,

Parte2: $Exc_1 \times C_v = -411.22$ US\$/año, (Comercialización)

Parte3: $Exc_1 \times (Tr + Di + PR + R) = -715.1$ US\$/año, (Peajes y cargos de red)

Parte4: $Exc_2 \times MC_m = 0$ US\$/año, Total Excedentes: -7205 US\$/año

La capacidad de exportación máxima declarada se ha fijado en $P_b^{max} = 400$ kW. En este sentido el cargo de respaldo anual es $backup = -24000.0$ USD/año.

El balance anual del AGPE fue negativo -31205.0 USD/año (importador neto).

$w_2=1$

Se obliga al modelo

$Exc_2 = 0$



Colombia

Ejemplo 6 - Autogeneración PV & BESS - Colombia

Universidad de los Andes

Facultad de Ingeniería

Educación Continua
Vicerrectoría Académica

¡ÓPTIMO!



	Sin PV (USD/año)	Con PV (USD/año)	Ahorro (USD/año)
Energía Activa	-182896.8	-7205.02	175691
Cargo de respaldo	0	-24000.05	-24000.0
Penalización por reactiva	-24942.4	0(*)	24942.4
Total	-207839.2	-31205.0	176480.0

Tabla 16: Costo energético de una industria genérica en Colombia con AGPE 569 kWp y BESS 2000 kWh, 2024

AHORRO de 84%

$w_2=1$

Se obliga al modelo

Exc2 = 0



Evaluación Económica

NPV

TIR

PBT

B/C Ratio


VP Sistema PV a 770 USD/kWp

VP Sistema BESS: 220 USD/kWh (batería)
88 USD/kW (inversor GFL)

(incluye costo inversor y accesorios)

Tasa de descuento 5%

Vida del proyecto 20 años





Ejemplo 6 - Autogeneración PV & BESS - Colombia

Código	Tecn.	¿Desp. óptimo?	VPN (Eur)	TIR (%)	PBT (años)	B/C	Conf.
M1.4.E4c	PV	no	1340644	32.5	3.4	4.1	Media
M1.4.E6c	PV+BESS	si	1286000	18.7	6.1	2.4	Alta

Tabla 17: Indicadores de bondad financiera del proyecto PV+BESS en Colombia, 2024

¡TIR solo PV mejor que TIR PV+BESS!

El costo de la batería (sigue muy alto) penaliza el proyecto.

No obstante tener batería mejora confiabilidad


¿cuanto cuesta la confiabilidad?

$w_2=1$
Se obliga al modelo
Exc2 = 0



Ejemplo 6 - Autogeneración PV & BESS - Colombia


DESPACHO OPTIMO



$w_2=1$
Se obliga al modelo
Exc2 = 0



DESPACHO OPTIMO




$w_2=1$
Se obliga al modelo
 $Exc2 = 0$



Ejemplo 6 - Autogeneración PV & BESS - Colombia

DESPACHO OPTIMO



$w_2=1$
Se obliga al modelo
 $Exc2 = 0$



DESPACHO OPTIMO



$w_2=1$
Se obliga al modelo
 $Exc2 = 0$

RESUMEN de PRUEBAS

Código	Tecn.	¿Desp. óptimo?	VPN (Eur)	TIR (%)	PBT (años)	B/C	Conf.
España	PV	no	394710	15.0	7.7	2.0	Media
Colombia	PV	no	1340644	32.5	3.4	4.1	Media
España	PV+BESS	si	339341	9.4	12	1.4	Alta
Colombia	PV+BESS	si	1286000	18.7	6.1	2.4	Alta

Tabla 18: Indicadores de bondad financiera del proyecto PV+BESS en España Colombia, 2024

Caso Colombia
 $w_2=1$
 Se obliga al modelo
 $Exc2 = 0$

ESPECIFICACION DEL SISTEMA PV/BESS

CRITERIOS:

- A) Sin BESS: Sólo crédito de energía,
- B) Sin BESS: Crédito + Excedentes horarios (Exc2)
- C) Con BESS: Solo crédito, Exc 2 = 0 (*Optimo*)
- D) con BESS: Crédito + Exc 2 > 0 (*Optimo*)**



¿que pasa cuando $w_2=0$?
Se obliga al modelo a
que $Exc2 > 0$

```
88      w2      /0/      ;|  
89      SOC.up(t) = ((1-DoD)/2+DoD)*C;
```

¿Convergerá?